

```

//Commande principale:// démarrer

pour nouveau
// Initialise les paramètres par défaut de l'écran,
// du crayon et de la tortue. //
eff fcfg noire efftxt déroule accélère cachetortue
lèvecrayon
fin

pour initialise
donne "anglex 0
donne "angley 0
donne "anglez 0
donne "m card :couleur
fin

pour cercle :r
ellipse [ xcoord ycoord :r :r ]
fin

pour dessine
tantque vrai [
donne "n 1
donne "anglex :anglex +5
donne "angley :angley +7
donne "anglez :anglez +3
répète :m [
donne "xd item :formex :n
donne "yd item :formey :n
donne "zd item :formez :n
donne "zx :xd *(cos :anglez) - :yd *(sin :anglez) - :xd
donne "zy :xd *(sin :anglez) + :yd *(cos :anglez) - :yd
donne "yx (:xd +:zx)*(cos :angley) - :zd *(sin :angley) - (:xd +:zx)
donne "yz (:xd +:zx)*(sin :angley) + :zd *(cos :angley) - :zd
donne "xy (:yd +:zy)*(cos :anglex) - (:zd +:yz)*(sin :anglex) - (:yd +:zy)
donne "xz (:yd +:zy)*(sin :anglex) + (:zd +:yz)*(cos :anglex) - (:zd +:yz)
donne "z :zd + (:xz +:yz)
donne "x (:xd + (:yx +:zx))*((:z +300)/295)
donne "y (:yd + (:zy +:xy))*((:z +300)/295)
fpos :x :y
fcc item :couleur :n cercle 4+(:z +200) /50 donne "n :n +1
]
attends 0,05
]
fin

pour forme3D :cas // :cas = 1, 2 ou 3 //
teste :cas = 1
sivrai [
// cube //
donne "formex [80 (-80) (-80) 80 80 (-80) (-80) 80]
donne "formey [80 80 (-80) (-80) (-80) (-80) 80 80]
donne "formez [80 80 80 80 (-80) (-80) (-80) (-80)]
donne "couleur [1000 10000 200 2000 30000 3000 600 6000]
]
sifaux [
teste :cas = 2
sivrai [
// tétraèdre //
donne "formex [0 (-80) 80 0]
donne "formey [-65 (-65) (-65) 65]
donne "formez [80 (-50) (-50) 0]
donne "couleur [10000 20000 30000 6000]
]
sifaux [
// diamant //
donne "formex [80 (-80) (-80) 80 0 0]
donne "formey [0 0 0 0 113 (-113)]
donne "formez [80 80 (-80) (-80) 0 0]
donne "couleur [2000 600 20000 6000 10000 1000]
]
]
]

```

```
    ]  
  ]  
fin
```

```
pour démarrer  
partage [ "anglex "angley "anglez "couleur "formex "formey "formez "m ]  
nouveau forme3D 1 // :cas = 1, 2 ou 3 //  
initialise dessine  
fin
```